

## NAG Library Routine Document

### F07GJF (DPPTRI)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F07GJF (DPPTRI) computes the inverse of a real symmetric positive definite matrix  $A$ , where  $A$  has been factorized by F07GDF (DPPTRF), using packed storage.

#### 2 Specification

```
SUBROUTINE F07GJF (UPLO, N, AP, INFO)
  INTEGER          N, INFO
  REAL (KIND=nag_wp) AP(*)
  CHARACTER(1)    UPLO
```

The routine may be called by its LAPACK name *dpptri*.

#### 3 Description

F07GJF (DPPTRI) is used to compute the inverse of a real symmetric positive definite matrix  $A$ , the routine must be preceded by a call to F07GDF (DPPTRF), which computes the Cholesky factorization of  $A$ , using packed storage.

If UPLO = 'U',  $A = U^T U$  and  $A^{-1}$  is computed by first inverting  $U$  and then forming  $(U^{-1})U^{-T}$ .

If UPLO = 'L',  $A = LL^T$  and  $A^{-1}$  is computed by first inverting  $L$  and then forming  $L^{-T}(L^{-1})$ .

#### 4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

#### 5 Arguments

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies how  $A$  has been factorized.  
 UPLO = 'U'  
 $A = U^T U$ , where  $U$  is upper triangular.  
 UPLO = 'L'  
 $A = LL^T$ , where  $L$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: AP(\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .  
*On entry:* the Cholesky factor of  $A$  stored in packed form, as returned by F07GDF (DPPTRF).

*On exit:* the factorization is overwritten by the  $n$  by  $n$  matrix  $A^{-1}$ .

More precisely,

if UPLO = 'U', the upper triangle of  $A^{-1}$  must be stored with element  $A_{ij}$  in AP( $i + j(j - 1)/2$ ) for  $i \leq j$ ;

if UPLO = 'L', the lower triangle of  $A^{-1}$  must be stored with element  $A_{ij}$  in AP( $i + (2n - j)(j - 1)/2$ ) for  $i \geq j$ .

4: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Diagonal element  $\langle value \rangle$  of the Cholesky factor is zero; the Cholesky factor is singular and the inverse of  $A$  cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where  $c(n)$  is a modest function of  $n$ ,  $\epsilon$  is the *machine precision* and  $\kappa_2(A)$  is the condition number of  $A$  defined by

$$\kappa_2(A) = \|A\|_2 \|A^{-1}\|_2.$$

## 8 Parallelism and Performance

F07GJF (DPPTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately  $\frac{2}{3}n^3$ .

The complex analogue of this routine is F07GWF (ZPPTRI).

## 10 Example

This example computes the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix}.$$

Here  $A$  is symmetric positive definite, stored in packed form, and must first be factorized by F07GDF (DPPTRF).

### 10.1 Program Text

```

Program f07gjfe

!      F07GJF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: dpptrf, dpptri, nag_wp, x04ccf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter      :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                  :: i, ifail, info, j, n
Character (1)            :: uplo
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: ap(:)
!      .. Executable Statements ..
Write (nout,*) 'F07GJF Example Program Results'
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (ap(n*(n+1)/2))

!      Read A from data file

Read (nin,*) uplo
If (uplo=='U') Then
  Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
Else If (uplo=='L') Then
  Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
End If

!      Factorize A
!      The NAG name equivalent of dpptrf is f07gdf
Call dpptrf(uplo,n,ap,info)

Write (nout,*)
Flush (nout)
If (info==0) Then

!      Compute inverse of A
!      The NAG name equivalent of dpptri is f07gjf
Call dpptri(uplo,n,ap,info)

!      Print inverse

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04ccf(uplo,'Nonunit',n,ap,'Inverse',ifail)

```

```

Else
  Write (nout,*) 'A is not positive definite'
End If

End Program f07gjfe

```

## 10.2 Program Data

```

F07GJF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  4.16
-3.12   5.03
  0.56  -0.83   0.76
-0.10   1.18   0.34   1.18   :End of matrix A

```

## 10.3 Program Results

F07GJF Example Program Results

```

Inverse
      1           2           3           4
1      0.6995
2      0.7769      1.4239
3      0.7508      1.8255      4.0688
4     -0.9340     -1.8841     -2.9342      3.4978

```

---